Konvexitás, inflexió

Ebben a fejezetben a függvénygörbe további alaki sajátosságaival ismerkedünk meg. A konvexitás-konkávitás lényeges információt hordoz a függvény növekedési ütemének változásáról.

A konvexitás, konkávitás és az inflexió fogalma

Tekintsük a következõ függvényt:

```
> restart:
> f:=x->(1-2^(-x+10))/(2*(1+2^(-x+10)))+1/2;
```

$$f := x \rightarrow \frac{1-2^{(-x+10)}}{2+2\,2^{(-x+10)}}+\frac{1}{2} \qquad (1.1)$$

```
> solve((D@@2)(f)(x));
```

$$10 \qquad (1.2)$$

```
> plot(f(x),x=0..20,color=blue,thickness=2);
```



A függvény szigorúan monoton növekvõ. Növekedésének üteme kb. az $x = 10$ értékig egyre nagyobb, innentõl viszont egyre lassulóan növekszik.

Figyeljük most a függvény és érintõinek viszonyát:

```
> erintok:=proc(f,a,b,n,k)
>    local i,t,rajz1,rajz2,pont;
>  pont:=plottools[point]([10,f(10)],color=cyan,symbol=circle):
> for i from 1 to n do
>    t:=a+(b-a)/n*(i-1);
> rajz1:=plot(f(x),x=a..b,color=blue,thickness=2):
> if i=1 or i=n  then
> rajz2:=plot(D(f)(t)*(x-t)+f(t),x=t-.5*k*(b-a)/n..t+0.5*k*(b-a)
  /n,color=red,thickness=2): else
> rajz2:=plot(D(f)(t)*(x-t)+f(t),x=t-(k+1)*(b-a)/n..t+(k+1)*(b-a)
  /n,color=red,thickness=2): fi:
>       kep||i:=plots[display]([rajz1,rajz2,pont]):
>    od;
```

```
> end:
> n:=20:k:=2.7:
> erintok(f,1,18,n,k):
> plots[display]([kep||(1..n)],title=`Figyeljük az érintõ és a függvény viszonyát`,insequence=true);
```




```
>
```

Amíg a függvény egyre gyorsabban növekszik, addig érintõje a függvény "alatt van", a lassuló növekedés intervallumán pedig az érintõk a "függvény fölött" helyezkednek el.

Definíció.

Ha az [a,b] intervallumon differenciálható f függvény bármely [a,b] -hez tartozó pontban húzott érintõje a görbe alatt (felett) helyezkedik el, akkor azt mondjuk, hogy a függvény görbéje az [a,b] intervallumban alulról konvex (konkáv).
Ha a függvény nem feltétlenül differenciálható az [a,b] számközön, akkor a következõképpen definiálhatjuk az elõzõ fogalmakat:
Ha az f függvény görbének megfelelõ darabja az [a,b] tetszõleges pontjaihoz tartozó szelõ alatt (felett) helyezkedik el, akkor azt mondjuk, hogy az f görbéje az [a,b] intervallumon alulról konvex (konkáv).

Definíció.
A konvex és konkáv ívdarabok találkozási pontját inflexiós pontnak nevezzük.

Szeretnénk elégséges feltételt nyerni a konvexitásra-konkávitásra és az inflexiós pont létezésére! Rajzoljuk föl az elõbbi f függvényt és elsõ és második deriváltját!

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=2..18,color=[blue,red,green],
  thickness=2,legend=["f","D(f)","(D@@2)(f)"]);
```




```
>
```

Azt látjuk, hogy amíg a függvény elsõ deriváltja növekvõ és ezzel együtt második deriváltja pozitív, addig a függvény alulról konvex. Azon az intervallumon pedig, ahol a derivált csökkenõ és egyidejûleg a második derivált negatív a függvény konkáv. Ez általában is így van, errõl szól a következõ szekció elsõ tétele.

A konvexitás vizsgálata

Tétel.

Ha az f függvény második deriváltja, $\frac{d^2}{dx^2} f(x)$ az [a,b] intervallumon pozitív (negatív), akkor a függvény görbéje az [a,b]-n alulról konvex (konkáv).

A helyi szélsõérték létezésére vonatkozó feltétellel analóg az inflexiós pont létezésére vonatkozó szükséges feltétel.

Tétel.

A legalább kétszer differenciálható f függvénynek csak olyan x_0 pontban **lehet inflexió**ja, ahol második deriváltja nulla, azaz

$$\frac{d^2}{dx^2} f\left(x_0\right) = 0 \qquad \text{(Másképp: f "}(x_0)\text{ =0, vagy } \left(\left(D^{(2)}\right)(f)\right)\left(x_0\right) = 0 \text{).}$$

Az inflexiós pont létezésére vonatkozó elégséges feltétel is analóg a helyi szélsõértékre vonatkozó tétellel, itt eggyel magasabb rendû derivált szerepel.

Tétel. (Az inflexiós pont létezésére vonatkozó elsõ elégséges feltétel.)

Ha $\frac{d^2}{dx^2} f\left(x_0\right) = 0$ **és** $\frac{d^2}{dx^2} f\left(x_0\right)$ az x_0 pontban elõjelet vált, akkor az f függvénynek az x_0 pontban inflexiója van.
Más jelöléssel:
Ha f " (x_0) = 0 és f " (x) az x_0 pontban elõjelet vált, akkor az f függvénynek az x_0 pontban inflexiója van.

Tétel. (Az inflexiós pont létezésére vonatkozó második elégséges feltétel.)

Ha $\frac{d^2}{dx^2} f\left(x_0\right) = 0$ és $\frac{d^3}{dx^3} f\left(x_0\right) \neq 0$, akkor f-nek az x_0 helyen inflexiós pontja van.
Szavakban kifejezve ha az x_0 helyen az f második deriváltja nulla és a harmadik deriváltja nem nulla, akkor f-nek az x_0 helyen inflexiós pontja van.

Példa.

Határozzuk meg az

$$f(x) = 2\,x^2 - x^3$$

függvény konvexitási intervallumait és inflexiós pontjának koordinátáit!

Megoldás.

Vegyük föl a függvényt!

```
> f:=x->2*x^2-x^3;
```

$$f := x \rightarrow 2\,x^2 - x^3 \qquad (2.1)$$

Képezzük a második deriváltat.

```
> Diff('f(x)',x$2)=(D@@2)(f)(x);
```

$$\frac{d^2}{dx^2} f(x) = 4 - 6\,x \tag{2.2}$$

A solve utasítással megoldjuk a $\frac{d^2}{dx^2} f(x) = 0$ egyenletet:

```
> zh:=[solve(rhs(%))];
```

$$zh := \left[\frac{2}{3}\right] \tag{2.3}$$

Ha ezen a helyen a harmadik derivált nem nulla, akkor valóban van inflexiós pont:

```
> '(D@@3)(f)'(zh[1])=((D@@3)(f)(zh[1]));
```

$$\left(\left(D^{(3)}\right)(f)\right)\left(\frac{2}{3}\right) = -6 \tag{2.4}$$

A harmadik derivált negatív a második derivált zérushelyén, tehát van inflexiós pont. Egyuttal azt is tudjuk, hogy a második derivált csökkenõ, tehát konvex ívet konkáv vált föl. A függvény és a második valamint a harmadik derivált:

```
> plot([f(x),D(f)(x),(D@@2)(f)(x),0],x=-1..2,-4..4,color=[blue,
  red,green,white],thickness=2,legend=["f","D(f)","(D@@2)(f)"," "]
  );
```



Az inflexiós pont második koordinátája:

```
> f(2/3);
```

$$\frac{16}{27} \tag{2.5}$$

Ennek közelítő értéke:

```
> evalf(%);
```

$$0.5925925926 \tag{2.6}$$

```
>
```

Kérdések, feladatok

Ellenõrzõ kérdések

1. Definiálja a konvexitás, konkávitás és az inflexiós pont fogalmát!
2. Milyen információt nyújt a differenciálható függvény viselkedésérõl a görbe konvexitása?
3. Mit mondhatunk a görbe konvexitásáról, ha deriváltja növekvõ (csökkenõ)?

4. Mely pontokban lehet a legalább kétszer deriválható függvénynek inflexiója?
5. Hogyan dönthetjük el, hogy egy adott pontban van -e inflexiós pont?

Gyakorló feladatok

Határozza meg a következõ függvények konvex és konkáv szakaszait és inflexiós pontjait:

1. $f(x) = 3x^2 - x^3$; 2. $f(x) = \sqrt{1 + x^2}$; 3. $f(x) = e^{(-x^2)}$; 4. $f(x) = \ln(1 + x^2)$; 5. $f(x) = \dfrac{1}{\ln(x)}$

6. $f(x) = x \sin(\ln(x))$.

2. Mutassa meg, hogy az $f(x) = \dfrac{x+1}{x^2+1}$ függvénynek három inflexiós pontja van és ezek egy egyenesen helyezkednek el!